

▼ 5. Mátrix sajátértékének meghatározása, hatványmódszer

Az ebben a fejezetben tárgyalt fogalmakat a mechanikai, kvantummechanikai, geometriai problémák megoldása során széles körűen felhasználják.

▼ 5. 1. Definíció

Az $A \in \mathbb{C}^{n \times n}$ négyzetes mátrix **sajátértéke** a λ valós vagy komplex szám, ha található legalább egy olyan v (oszlop) vektor, melyre

$$Av = \lambda v, \quad v \neq 0.$$

A v (általában komplex koordinátás) v vektor az A mátrix λ -hoz tartozó **sajátvektora**. $\square$

▼ 5. 2. Definíció

Egy mátrix sajátértékeinek összességét a mátrix **spektrumának**, a sajátértékeket **spektrálparamétereknek** is nevezik. $\square$

Az 5.1. definíció ekvivalens az

$$(A - \lambda E)v = 0$$

egyenlettel, ahol E az $n \times n$ -es egységmátrix. Az így kapott mátrixegyenlet viszont egy olyan **homogén** (amelyben minden, a jobb oldalon álló konstans értéke 0) **lineáris egyenletrendszer** mátrixszorzatos alakja, amelynek együtttható mátrixa az $A - \lambda E$ mátrix, ismeretlenei pedig a v koordinátái.

Mivel a sajátvektornak a definíció szerint legalább egy koordinátája nem 0, ezért az egyenletrendszer **nem** lehet **szabályos**, vagyis nem lehet egyértelmű megoldása, hiszen a homogén, szabályos lineáris egyenletrendszer megoldása csak a zérusvektor lehet.

Az ugyanannyi egyenletet, mint ahány ismeretlent tartalmazó, nem szabályos lineáris egyenletrendszer **együttthatómátrixa determinánsának 0** értékűnek kell lennie.

▼ 5. 3. Tétel

Adott λ -hoz tartozó sajátvektor létezésének szükséges és elégséges feltétele:

$$\det(A - \lambda E) = 0. \quad \square$$

Tehát azokhoz a λ -khoz tartozik sajátvektor, melyek eleget tesznek az 5.3. tétel feltételének.

Más szóval: Az A mátrix **sajátértékei** a $\det(A - \lambda E) = 0$ egyenlet **megoldásai**.

▼ 5. 4. Definíció

A $\det(A - \lambda E) = 0$ egyenlet bal oldalát az A mátrix **karakterisztikus polinomjának**, az egyenletet pedig az A mátrix **karakterisztikus egyenletének** nevezzük. $\square$

Ha megtaláltuk az A mátrix valamelyik sajátértékét (λ_0) az ehhez tartozó sajátvektort úgy számítjuk ki, hogy megoldjuk a $(A - \lambda_0 E)v = 0$ homogén lineáris egyenlet rendszert.

▼ 5. 5. Definíció

A λ sajátértékhez tartozó maximális számú lineárisan független sajátvektorok számát a **sajátérték geometriai multiplicitásának** nevezzük. $\square$

Egy sajátérték **algebrai multiplicitásán** azt értjük, hogy hányszoros multiplicitású gyöke a karakterisztikus egyenletnek. $\square$

▼ 5. 6. A sajátérték és a sajátvektor meghatározása Maple-lel

Használt Maple parancsok: [factor, rhs, lhs](#), [Eigenvalues](#), [Eigenvectors](#), [DotProduct](#), [pointplot](#)

1. Példa Adjuk meg az $A := \begin{bmatrix} -2 & 2 & -3 \\ 2 & 1 & -6 \\ -1 & -2 & 0 \end{bmatrix}$ mátrixának sajátértékeit és sajátvektorait a Maple

beépített parancsaival

Megoldás

> `restart; with(LinearAlgebra) : A := Matrix([[-2, 2, -3], [2, 1, -6], [-1, -2, 0]]) :`

Ha csak a sajátértéket keressük, használjuk az *Eigenvalues* parancsot.

> `sajatertek := Eigenvalues(A)`

$$sajatertek := \begin{bmatrix} 5 \\ -3 \\ -3 \end{bmatrix}$$

A sajátértékek, azok multiplicitása és a hozzájuk tartozó sajátvektorok az *Eigenvectors* paranccsal együtt is megkaphatók.

> `s := Eigenvectors(A)`

$$s := \begin{bmatrix} -3 \\ -3 \\ 5 \end{bmatrix}, \begin{bmatrix} 3 & -2 & -1 \\ 0 & 1 & -2 \\ 1 & 0 & 1 \end{bmatrix}$$

Az első oszlopvektor a sajátértékeket, a második mátrix oszlopai az egyes sajátértékekhez tartozó sajátvektorokat tartalmazzák. Olvassuk ki a sajátértékeket és a sajátvektorokat!

> `sajatertek1 := s[1][1]; sajatertek2 := s[1][2]; sajatertek3 := s[1][3]`

$$sajatertek1 := -3$$

$$sajatertek2 := -3$$

$$sajatertek3 := 5$$

> `sajatvektor1 := s[2][1..3, 1]; sajatvektor2 := s[2][1..3, 2]; sajatvektor3 := s[2][1..3, 3];`

$$sajatvektor1 := \begin{bmatrix} 3 \\ 0 \\ 1 \end{bmatrix}$$

$$sajatvektor2 := \begin{bmatrix} -2 \\ 1 \\ 0 \end{bmatrix}$$

$$\text{sajátvektor3} := \begin{bmatrix} -1 \\ -2 \\ 1 \end{bmatrix}$$

▼ 5. 7. A sajátértékek eloszlása

▼ 5. 7. 1. Tétel

Egy $\mathbf{A}$ mátrix sajátértékeinek abszolútértékét az $\mathbf{A}$ valamennyi normája felülről becsli. $\square$

Bizonyítás

A sajátérték definíciója alapján

$$|\lambda| \|\mathbf{v}\| = |\lambda \cdot \mathbf{v}| = \|\mathbf{A} \cdot \mathbf{v}\| \leq \|\mathbf{A}\| \|\mathbf{v}\|,$$

amiből $\|\mathbf{v}\| \neq \mathbf{0}$ -val osztva $|\lambda| \leq \|\mathbf{A}\|$. (Ez az egyenlőtlenség csak akkor igaz, ha a használt mátrix- és vektornorma kompatibilis.)

2. Példa Adjunk felső becslést az $\mathbf{A} = \begin{bmatrix} -2 & 2 & -3 \\ 2 & 1 & -6 \\ -1 & -2 & 0 \end{bmatrix}$ mátrix sajátértékeire a 3. fejezetben

megismert normák segítségével.

Megoldás

- > `restart; with(LinearAlgebra) : A := Matrix([[-2, 2, -3], [2, 1, -6], [-1, -2, 0]]) :`
- > `sor_osszeg_norma := MatrixNorm(A, 1); oszlop_osszeg_norma := MatrixNorm(A, infinity);`
 $\text{sor_osszeg_norma} := 9$
 $\text{oszlop_osszeg_norma} := 9$
- > `Eigenvalues(A)`

$$\begin{bmatrix} 5 \\ -3 \\ -3 \end{bmatrix}$$

Valóban, minden sajátérték abszolútértéke kisebb, mint 9.

▼ 5. 7. 2. Tétel

Gersgorin tétele: Az $\mathbf{A}$ valós vagy komplex elemű mátrix valamennyi sajátértéke a komplex számsík alábbi zárt köreinek egyesítésében van:

$$|z - a_{ii}| \leq r_i,$$

ahol

$$r_i = \sum_{j=1}^n |a_{ij}| \quad (i=1,2, \dots, n, j \neq i). \quad \square$$

Bizonyítás

Legyen λ sajátértékhez tartozó v sajátvektor legnagyobb abszolútértékű komponensének indexe i . Az $Av = \lambda v$ sajátértékegyenlet i -edik komponense x_i .

$$\sum_{j=1}^n a_{ij} x_j = \lambda x_i$$

x_i a bal és a jobb oldalon is szerepel, vigyük át a jobb oldalra

$$\sum_{j=1}^n a_{ij} x_j = (\lambda - a_{ii}) x_i$$

x_i -vel osztva, majd abszolútértékre áttérve adódik, hogy

$$|\lambda - a_{ii}| \leq \sum_{j=1, j \neq i}^n |a_{ij}| = r_i,$$

amivel a tételt beláttuk.

3. Példa Írjuk fel és szemléltessük a 2. példa Gersgorin köreit

Megoldás

> $A := \text{Matrix}([[-2, 2, -3], [2, 1, -6], [-1, -2, 0]])$:

> **for** i **to** 3 **do**

$r[i] := 0$;

for j **to** 3 **do** **if** $i \neq j$ **then** $r[i] := r[i] + |A_{i,j}|$ **else** $r[i] := r[i]$ **end if** **end do**;

$kor[i] := (x - A_{i,i})^2 + y^2 = r_i^2$;

$\text{print}(kor[i])$

end do :

$$(x + 2)^2 + y^2 = 25$$

$$(x - 1)^2 + y^2 = 64$$


$$x^2 + y^2 = 9$$

> $\text{with}(\text{plots})$:

$abra1 := \text{implicitplot}(\{kor[1], kor[2], kor[3]\}, x = -10..10, y = -10..10, \text{scaling} = \text{unconstrained}, \text{color} = \text{blue})$:

$abra2 := \text{pointplot}(\{[Eigenvalues(A)_1, 0], [Eigenvalues(A)_2, 0], [Eigenvalues(A)_3, 0]\}, \text{color} = \text{red}, \text{symbolsize} = 30, \text{symbol} = \text{diagonalcross})$:

$\text{display}([abra1, abra2])$



Az ábrán a köröket és a sajátértékeket ábrázoltuk, mindkét sajátérték a körök uniójának halmazában helyezkedik el.

▼ 5. 8. A legnagyobb sajátérték numerikus meghatározása hatványmódszerrel

Használt Maple parancsok: [convert](#) [Vector](#)

A sajátértékek meghatározásakor kapott karakterisztikus polinom fokszáma a mátrix nagyságától függ. Egy 3x3-as mátrix karakterisztikus polinomja is harmadfokú egyenlet megoldásához vezet. Ezért van nagy szerepe a sajátértékek meghatározásánál a numerikus módszereknek.

Tegyük fel, hogy

- 1; az A $n \times n$ -es mátrix sajátértékei valós számok, és $|\lambda_1| > |\lambda_2| \geq |\lambda_3| \geq \dots \geq |\lambda_n|$
- 2; az A sajátvektorai $v_1, v_2, v_3, \dots, v_n$ legyenek lineárisan függetlenek, azaz egyik se legyen előállítható a többi lineáris kombinációjaként.
- 3; x_0 tetszőleges, a sajátvektor 0. közelítésének tekintett vektor, amely felírható a sajátvektorok lineáris kombinációjaként úgy, hogy

$$x_0 = c_1 v_1 + c_2 v_2 + \dots + c_n v_n$$

Ezt az egyenletet az A mátrixszal többször megszorozva, valamint felhasználva a $Av = \lambda v$, az előző egyenlet a következő alakú lesz

$$\begin{aligned}
\mathbf{A}\mathbf{x}_0 &= c_1\lambda_1\mathbf{v}_1 + c_2\lambda_2\mathbf{v}_2 + \dots + c_n\lambda_n\mathbf{v}_n \\
\mathbf{A}^2\mathbf{x}_0 &= c_1\lambda_1^2\mathbf{v}_1 + c_2\lambda_2^2\mathbf{v}_2 + \dots + c_n\lambda_n^2\mathbf{v}_n \\
&\vdots \\
&\vdots \\
&\vdots \\
\mathbf{A}^n\mathbf{x}_0 &= c_1\lambda_1^n\mathbf{v}_1 + c_2\lambda_2^n\mathbf{v}_2 + \dots + c_n\lambda_n^n\mathbf{v}_n \\
\mathbf{A}^{n+1}\mathbf{x}_0 &= c_1\lambda_1^{n+1}\mathbf{v}_1 + c_2\lambda_2^{n+1}\mathbf{v}_2 + \dots + c_n\lambda_n^{n+1}\mathbf{v}_n
\end{aligned}$$

Az utolsó két egyenlet mindkét oldalát a legnagyobb sajátérték n -edik és $n+1$ -edik hatványával elosztva

$$\begin{aligned}
\frac{1}{\lambda_1^n} \mathbf{A}^n \mathbf{x}_0 &= c_1 \mathbf{v}_1 + c_2 \left(\frac{\lambda_2}{\lambda_1} \right)^n \mathbf{v}_2 + \dots + c_n \left(\frac{\lambda_n}{\lambda_1} \right)^n \mathbf{v}_n \approx c_1 \mathbf{v}_1 \\
\frac{1}{\lambda_1^{n+1}} \mathbf{A}^{n+1} \mathbf{x}_0 &\approx c_1 \mathbf{v}_1
\end{aligned}$$

A közelítésnél felhasználtuk, ha $n \gg 1$, a törtes kifejezések sokkal kisebbek 1-nél, így elhagyhatóak. A most kapott egyenletek mindkét oldalát szorozzuk meg egy tetszőleges $\mathbf{y}$ oszlopvektorral, melyre $\mathbf{x}_0 \mathbf{y} \neq 0$, így már a két egyenletet el lehet osztani egymással.

$$\begin{aligned}
\frac{1}{\lambda_1^n} \mathbf{A}^n \mathbf{x}_0 \mathbf{y} &\approx c_1 \mathbf{v}_1 \mathbf{y} \\
\frac{1}{\lambda_1^{n+1}} \mathbf{A}^{n+1} \mathbf{x}_0 \mathbf{y} &\approx c_1 \mathbf{v}_1 \mathbf{y} \\
\lambda_1 &\approx \frac{\mathbf{A}^{n+1} \mathbf{x}_0 \mathbf{y}}{\mathbf{A}^n \mathbf{x}_0 \mathbf{y}}
\end{aligned}$$

Az eljárást addig folytatjuk, amíg λ_1 két egymás után következő közelítésének eltérése egy előre megadott hibánál kisebb nem lesz.

4. Példa Határozzuk meg az $\mathbf{A} = \begin{bmatrix} 5 & -4 & -2 & 4 \\ 3 & -2 & 0 & 2 \\ 0 & 0 & 1 & 1 \\ 0 & 0 & 1 & 1 \end{bmatrix}$ mátrix legnagyobb abszolútértékű sajátértékét

hatványmódszerrel.

Megoldás

> *restart; with(LinearAlgebra) :*

A := Matrix(4, 4, [[5, -4, -2, 4], [3, -2, 0, 2], [0, 0, 1, 1], [0, 0, 1, 1]]) : sajátértékek
:= Eigenvalues(A);

$$\mathbf{A} := \begin{bmatrix} 5 & -4 & -2 & 4 \\ 3 & -2 & 0 & 2 \\ 0 & 0 & 1 & 1 \\ 0 & 0 & 1 & 1 \end{bmatrix}$$

$$sajatertekek := \begin{bmatrix} 0 \\ 1 \\ 2 \\ 2 \end{bmatrix}$$

A kiindulási x_0 vektor legyen $[1,0,0,0]$, a tetszőleges y vektor $[1,1,1,1]$. (A mátrixból a *convert* eljárással tudunk vektort előállítani, hogy azonos típusu objektumokkal végezzük a műveleteket, majd két vektor skáláris szorzatát kell venni.) Az első közelítéshez az $\frac{A^2 x_0 y}{Ax_0 y}$ hányadost kell képezni.

> $x0 := Matrix(4, 1, [1, 0, 0, 0])$; $y := Vector([1, 1, 1, 1])$;
 > $Ax0 := convert(A.x0, Vector)$; $nevezo := DotProduct(Ax0, y)$;

$$Ax0 := \begin{bmatrix} 5 \\ 3 \\ 0 \\ 0 \end{bmatrix}$$

$$nevezo := 8$$

> $A2x0 := convert((A^2).x0, Vector)$; $samlalo := DotProduct(A2x0, y)$

$$A2x0 := \begin{bmatrix} 13 \\ 9 \\ 0 \\ 0 \end{bmatrix}$$

$$samlalo := 22$$

Így az első közelítés

> $max_sajatertek_1 := \frac{samlalo}{nevezo}$

$$max_sajatertek_1 := \frac{11}{4}$$

Írjunk rövid programot, ami a közelítéseket addig végzi el, amíg az egymást követő két sajátérték-közelítés eltérése nem lesz egy adott értéknél kisebb!

	Megvalósítás Maple-ben megadott hibahatárig
Input : $x_0, y, A, \varepsilon (\varepsilon > 0)$. $n=1, \lambda_{\max, n} = \frac{A^{n+1} x_0 y}{A^n x_0 y}$ $n = n + 1$ amíg $ \lambda_{\max, n} - \lambda_{\max, n-1} < \varepsilon$	$A, x_0, y, k, hiba$ > $k := 100$; $hiba := 100$; > for i to k while $10^{-5} < hiba$ do $s_max 0 := 100$; $Ax0 i := convert(A^i.x0, Vector)$; $Ax0 (i + 1) := convert(A^{i+1}.x0, Vector)$; $s_max i := evalf$

Output : $\lambda_{\max}$	<pre> ($\frac{DotProduct(Ax0 (i + 1), y)}{DotProduct(Ax0 i, y)}$); hiba := evalf(s_max i - s_max (i - 1)) end do : > print(lépésszám:, i - 1); print(`hiba:` , hiba) ; print(maximális sajátérték:, s_max (i - 1)) lépésszám:, 17 hiba:, 0.000006539 maximális sajátérték:, 2.000006540 </pre>
---	---

▼ 5. 9. Geometriai alkalmazás: főtengetytranszformáció

Használt Maple parancsok: [arrow](#)

A következő geometriai alkalmazást a mechanika sok helyen használja. Felhasználjuk a szimmetrikus mátrixokra megfogalmazott következő tételt.

▼ 5. 9. 1. Tétel

Főtengelytétel: Minden $n \times n$ típusú, szimmetrikus mátrixnak van n darab, páronként egymásra merőleges sajátvektora. $\square$

5. Példa Szemléltessük a főtengelytételt a $\begin{bmatrix} 8 & 2 & 0 \\ 2 & 5 & 0 \\ 0 & 0 & 1 \end{bmatrix}$ mátrixon!

> restart; with(LinearAlgebra) : A := Matrix([[8, 2, 0], [2, 5, 0], [-0, 0, 1]]) :
 > s := Eigenvectors(A)

$$s := \begin{bmatrix} 1 \\ 4 \\ 9 \end{bmatrix}, \begin{bmatrix} 0 & -\frac{1}{2} & 2 \\ 0 & 1 & 1 \\ 1 & 0 & 0 \end{bmatrix}$$

Olvassuk ki a sajátvektorokat, majd a merőlegesség vizsgálatához képezzük a vektorok skaláris szorzatát a *DotProduct* parancs segítségével, a skaláris szorzat 0, a két vektor merőleges..

> sajátvektor1 := s[2][1..3, 1] : sajátvektor2 := s[2][1..3, 2] : sajátvektor3 := s[2][1..3, 3] :
 > DotProduct(sajátvektor1, sajátvektor2); DotProduct(sajátvektor1, sajátvektor3);
 DotProduct(sajátvektor3, sajátvektor2)
 0
 0
 0

▼ 5. 9. 2. Definíció

Változók homogén másodfokú kifejezését **kvadratikus (másodfokú) alaknak** nevezzük. $\square$

Az x, y változók kvadratikus alakja $a_{1,1} x^2 + 2a_{1,2} xy + a_{2,2} y^2$

Az x, y, z változók kvadratikus alakja

$$a_{1,1} x^2 + 2a_{1,2} xy + 2a_{1,3} xz + a_{2,2} y^2 + 2a_{2,3} yz + a_{3,3} z^2$$

▼ 5.9.3. Definíció

Az olyan kvadratikus alakot, amelyben nincs vegyes tag, **kanonikus alak**nak nevezzük. □

Az x, y, z változók kanonikus alakja $a_{1,1} x^2 + a_{2,2} y^2 + a_{3,3} z^2$

▼ 5.9.4. Tétel

Legyen $\mathbf{A}$ egy $n \times n$ -es szimmetrikus mátrix, $\mathbf{r}$ n dimenziós helyvektor (sorvektor). A kvadratikus alak az $\mathbf{A}$ szimmetrikus mátrix és az $\mathbf{r}$ segítségével $\mathbf{r} \mathbf{A} \mathbf{r}^T$ alakban írható fel. □

Bizonyítás $n=3$ esetre

> $A := \text{Matrix}(3, 3, [a_{1,1}, a_{1,2}, a_{1,3}, a_{1,2}, a_{2,2}, a_{2,3}, a_{1,3}, a_{2,3}, a_{3,3}]);$
 $r := \text{Vector}[\text{row}]([x, y, z])$

$$A := \begin{bmatrix} a_{1,1} & a_{1,2} & a_{1,3} \\ a_{1,2} & a_{2,2} & a_{2,3} \\ a_{1,3} & a_{2,3} & a_{3,3} \end{bmatrix}$$

$$r := \begin{bmatrix} x & y & z \end{bmatrix}$$

> $\text{kvadratikus_alak} := \text{simplify}(r.A.\text{Transpose}(r))$

$$\text{kvadratikus_alak} := x^2 a_{1,1} + 2xy a_{1,2} + 2xz a_{1,3} + y^2 a_{2,2} + 2yz a_{2,3} + z^2 a_{3,3}$$

A kvadratikus alakban szereplő x, y, z változók az $\mathbf{r}$ vektor $\mathbf{e}_1, \mathbf{e}_2, \mathbf{e}_3$ triviális bázisra vonatkoztatott koordinátái. Ha más bázist választunk, a kvadratikus alak formailag más alakú lesz. Legyenek az $\mathbf{A}$ mátrix sajátértékei $\lambda_1, \lambda_2, \lambda_3$, páronként egymásra merőleges, egységnyi hosszúságú sajátvektorai $\mathbf{s}_1, \mathbf{s}_2, \mathbf{s}_3$. Forgassuk el a koordinátarendszert úgy, hogy tengelyei legyenek párhuzamosak a sajátvektorokkal. Ebben az új koordinátarendszerben legyen

$$\mathbf{r} = u \mathbf{s}_1 + v \mathbf{s}_2 + w \mathbf{s}_3$$

Helyettesítsük ezt be az $\mathbf{r} \mathbf{A} \mathbf{r}^T$ kifejezésbe. Tekintettel arra, hogy

$$\mathbf{A} \mathbf{s}_1 = \lambda_1 \mathbf{s}_1, \quad \mathbf{A} \mathbf{s}_2 = \lambda_2 \mathbf{s}_2, \quad \mathbf{A} \mathbf{s}_3 = \lambda_3 \mathbf{s}_3$$

$$\mathbf{r} \mathbf{A} \mathbf{r}^T = (u \mathbf{s}_1 + v \mathbf{s}_2 + w \mathbf{s}_3) (u \lambda_1 \mathbf{s}_1 + v \lambda_2 \mathbf{s}_2 + w \lambda_3 \mathbf{s}_3) = u^2 \lambda_1 + v^2 \lambda_2 + w^2 \lambda_3$$

Itt felhasználtuk a sajátvektorok merőlegességét. Így az új koordinátarendszerben (u, v, w változókkal) a kvadratikus alak csak tiszta négyzetes tagokat tartalmaz, ezért formailag egyszerűbb. A koordinátarendszer ilyen megválasztását **főtengelytranszformációnak** nevezzük. A kapott kifejezés **kanonikus alak**.

6. Példa Tekintsük a $Q = 8x^2 - 4xy + 5y^2$ kvadratikus alakot. Adjuk meg a kifejezés kanonikus alakját és döntsük el, milyen görbét ír le ez az alak, ha a kifejezés értéke 1? Ábrázoljuk közös koordinátarendszerben a görbét és az egységnyi hosszúságú sajátvektorokat.

Megoldás

> $A := \text{Matrix}(2, 2, [8, -2, -2, 5]); s := \text{Eigenvectors}(A)$

$$A := \begin{bmatrix} 8 & -2 \\ -2 & 5 \end{bmatrix}$$

$$s := \begin{bmatrix} 4 \\ 9 \end{bmatrix}, \begin{bmatrix} \frac{1}{2} & -2 \\ 1 & 1 \end{bmatrix}$$

Tehát a görbe kanonikus alakja:

> $Q_kanonikus := s[1][1] u^2 + s[1][2](A) v^2$
 $Q_kanonikus := 4 u^2 + 9 v^2$

Az egységnyi hosszúságú sajátvektort úgy kapjuk, hogy a vektort elosztjuk a hosszával.

> $sajatvektor1 := \frac{s[2][1]..2, 1]}{\text{Norm}(s[2][1]..2, 1], 2)}; sajatvektor2 := \frac{s[2][1]..2, 2]}{\text{Norm}(s[2][1]..2, 2], 2)};$

$$sajatvektor1 := \begin{bmatrix} \frac{1}{5} \sqrt{5} \\ \frac{2}{5} \sqrt{5} \end{bmatrix}$$

$$sajatvektor2 := \begin{bmatrix} -\frac{2}{5} \sqrt{5} \\ \frac{1}{5} \sqrt{5} \end{bmatrix}$$

Rajzoljuk fel közös koordinátarendszerben az egységnyi hosszú sajátvektorokat és azt görbét, amit akkor kapunk, ha a kvadratikus alakot 1-gyel tesszük egyenlővé !

> $\text{with}(plots) :$


> $a1 := \text{arrow}(sajatvektor1, \text{shape} = \text{arrow}, \text{length} = 0.5, \text{color} = \text{blue}, \text{thickness} = 4) :$

> $a2 := \text{arrow}(sajatvektor2, \text{shape} = \text{arrow}, \text{length} = 0.5, \text{color} = \text{blue}, \text{thickness} = 4) :$

> $a3 := \text{implicitplot}(8 x^2 - 4 x y + 5 y^2 = 1, x = -0.5 .. 0.5, y = -0.5 .. 0.5) :$

> $\text{display}([a1, a2, a3], \text{scaling} = \text{constrained})$

>



Tehát a görbe ellipszis, melynek nagy és kistengelye a sajátvektorokkal párhuzamos irányú.

7. Példa Transzformáljuk kanonikus alakra a $Q = 2x^2 + 6y^2 + 8xz + 2z^2$ kvadratikus alakot, majd ábrázoljuk a $Q=1$ felületet és a sajátvektorokat.

> $A := \text{Matrix}(3, 3, [2, 0, 4, 0, 6, 0, 4, 0, 2]); s := \text{Eigenvectors}(A)$

$$A := \begin{bmatrix} 2 & 0 & 4 \\ 0 & 6 & 0 \\ 4 & 0 & 2 \end{bmatrix}$$

$$s := \begin{bmatrix} 6 \\ 6 \\ -2 \end{bmatrix}, \begin{bmatrix} 1 & 0 & -1 \\ 0 & 1 & 0 \\ 1 & 0 & 1 \end{bmatrix}$$

> $Q_kanonikus := \text{Eigenvalues}(A)[1] u^2 + \text{Eigenvalues}(A)[2] v^2 + \text{Eigenvalues}(A)[3] w^2$

> $sajatvektor1 := \frac{s[2][1..3, 1]}{\text{Norm}(s[2][1..3, 1], 2)}$; $sajatvektor2 := \frac{s[2][1..3, 2]}{\text{Norm}(s[2][1..3, 2], 2)}$;

$sajatvektor3 := \frac{s[2][1..3, 3]}{\text{Norm}(s[2][1..3, 3], 2)}$;


$$Q_kanonikus := -2 u^2 + 6 v^2 + 6 w^2$$

$$sajatvektor1 := \begin{bmatrix} \frac{1}{2} \sqrt{2} \\ 0 \\ \frac{1}{2} \sqrt{2} \end{bmatrix}$$

$$sajatvektor2 := \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}$$

$$sajatvektor3 := \begin{bmatrix} -\frac{1}{2} \sqrt{2} \\ 0 \\ \frac{1}{2} \sqrt{2} \end{bmatrix}$$

- > *a1 := arrow(sajatvektor1, length=2, shape=arrow, color=blue, thickness=4) :*
- > *a2 := arrow(sajatvektor2, length=2, shape=arrow, color=blue, thickness=4) :*
- > *a4 := arrow(sajatvektor3, length=2, shape=arrow, color=blue, thickness=4) :*
- > *a3 := implicitplot3d(2 x² + 6 y² + 8 x z + 2 z² = 1, x=-2 ..2, y=-2 ..2, z=-2 ..2, style*
= hidden, axes = normal) :
- > *display([a1, a2, a3, a4], scaling=constrained, grid=[150, 150, 150])*



A felület egyköpenyű hiperboloid.

▼ 5. 10. Fizikai alkalmazás


Használt Maple utasítások: [map](#), [assume](#), [Re](#), [Im](#), [dsolve](#)

A főtengetytranszformáció is tekinthető fizikai alkalmazásnak, a következő példa egy alap mechanikai problémát tárgyal.

8. Példa Két tömegpont csatolt rezgését a

$$m_1 \frac{d^2 x_1}{dt^2} = -k_1 x_1 + k(x_2 - x_1) \quad , \quad m_2 \frac{d^2 x_2}{dt^2} = -k_2 x_2 + k(x_1 - x_2)$$

differentiálegyenletrendszer írja le, ahol m_1, m_2 a rezgő tömegpont tömege, k_1, k_2 az egyes tömegpontokra ható rugóállandók, k a két tömegpont kölcsönhatása során fellépő erőt jellemző rugóállandó, x_1 és x_2 kitérés-idő függvényei. Ha $k=0$, akkor a két tömegpont között nincs kölcsönhatás. Az egyes tömegpontok mozgása egymástól független. Ha k értékét növeljük, akkor a tömegpontok egyre inkább befolyásolják egymás mozgását. Természetesen a különböző kezdeti feltételek igen eltérő mozgást eredményezhetnek.



Megoldás

> *restart; with(LinearAlgebra) :*

$$> de1 := \frac{d^2}{dt^2} x1(t) = -3 x1(t) + 2 (x2(t) - x1(t)) : de2 := \frac{d^2}{dt^2} x2(t) = 2 \cdot (x1(t) - x2(t)) :$$

Keressük meg a megoldásokat úgy, hogy átírjuk mátrix és vektor felhasználásával a differenciálegyenletrendszert lineáris rendszer alakba.

> $X := \text{Vector}([x1(t), x2(t)])$

$$X := \begin{bmatrix} x1(t) \\ x2(t) \end{bmatrix}$$

> $d2X := \text{Vector}(\text{map}(\text{diff}, X, t\$2))$

$$d2X := \begin{bmatrix} \frac{d^2}{dt^2} x1(t) \\ \frac{d^2}{dt^2} x2(t) \end{bmatrix}$$

> $A := \text{GenerateMatrix}([rhs(de1), rhs(de2)], [x1(t), x2(t)])[1]$

$$A := \begin{bmatrix} -5 & 2 \\ 2 & -2 \end{bmatrix}$$

> $rendszer := d2X = AX$

$$rendszer := \begin{bmatrix} \frac{d^2}{dt^2} x1(t) \\ \frac{d^2}{dt^2} x2(t) \end{bmatrix} = \begin{bmatrix} -5 x1(t) + 2 x2(t) \\ 2 x1(t) - 2 x2(t) \end{bmatrix}$$

Keressük a differenciálegyenletrendszer megoldását exponenciális alakban!

> $\text{subs}([x1(t) = e^{\omega t} t1, x2(t) = e^{\omega t} t2], \text{rendszer})$

$$\begin{bmatrix} \omega^2 e^{\omega t} t1 \\ \omega^2 e^{\omega t} t2 \end{bmatrix} = \begin{bmatrix} -5 e^{\omega t} t1 + 2 e^{\omega t} t2 \\ 2 e^{\omega t} t1 - 2 e^{\omega t} t2 \end{bmatrix}$$

$e^{\omega t}$ -vel osztva

> $\text{lhs}\left(\frac{\%}{e^{\omega t}}\right) = \text{simplify}\left(\text{rhs}\left(\frac{\%}{e^{\omega t}}\right)\right)$

$$\begin{bmatrix} \omega^2 t1 \\ \omega^2 t2 \end{bmatrix} = \begin{bmatrix} -5 t1 + 2 t2 \\ 2 t1 - 2 t2 \end{bmatrix}$$

Kaptunk egy sajátérték problémát ω -ra és sajátvektor problémát $t1, t2$ -re, hiszen a kapott egyenlet ekvivalens az

$$\omega^2 \begin{bmatrix} t1 \\ t2 \end{bmatrix} = \begin{bmatrix} -5 & 2 \\ 2 & -2 \end{bmatrix} \begin{bmatrix} t1 \\ t2 \end{bmatrix}$$

egyenlettel. ω^2 a sajátérték, $\begin{bmatrix} t1 \\ t2 \end{bmatrix}$ a sajátvektor. Adjuk meg az együtthatómátrix sajátértékeit és sajátvektorait.

> $s := \text{Eigenvectors}(A)$

$$s := \begin{bmatrix} -1 \\ -6 \end{bmatrix}, \begin{bmatrix} \frac{1}{2} & -2 \\ 1 & 1 \end{bmatrix}$$

Mivel mindkét sajátérték negatív, ezért ω értékei komplex számok.

> $e1 := \omega^2 = s_{1,1} : \omega[1] := \text{solve}(e1, \omega); t1[1] := s[2][1][1]; t2[1] := s[2][2][1]$

$$\omega_1 := I, -I$$

$$t1_1 := \frac{1}{2}$$

$$t2_1 := 1$$

> $e2 := \omega^2 = s[1][2] : \omega[2] := \text{solve}(e2, \omega); t1[2] := s[2][1][2]; t2[2] := s[2][2][2]$

$$\omega_2 := I\sqrt{6}, -I\sqrt{6}$$

$$t1_2 := -2$$

$$t2_2 := 1$$

Így a megoldások komplex értékű függvények lesznek

> $x1(t) = e^{\omega[1]_1 t} t1[1], x2(t) = e^{\omega[1]_1 t} t2[1]$

$$x1(t) = \frac{1}{2} e^{It}, x2(t) = e^{It}$$

$$> x1(t) = e^{\omega[2]_1 t} t1[2], x2(t) = e^{\omega[2]_1 t} t2[2]$$

$$x1(t) = -2 e^{I\sqrt{6} t}, x2(t) = e^{I\sqrt{6} t}$$

A differenciálegyenletrendszer valós megoldását ezek segítségével megkaphatjuk. Vegyük a komplex értékű függvények valós és képzetes részeit, tételezzük fel, hogy t valós értékű. (Hiszen létező fizikai problémát tárgyalunk.)

> *assume(t :: real)*

$$> \text{valos1} := x1(t) = \text{Re}\left(\text{evalc}\left(e^{\omega[1]_1 t} t1[1]\right)\right), x2(t) = \text{Re}\left(\text{evalc}\left(e^{\omega[1]_1 t} t2[1]\right)\right)$$

$$> \text{kepzetes1} := x1(t) = \text{Im}\left(\text{evalc}\left(e^{\omega[1]_2 t} t1[1]\right)\right), x2(t) = \text{Im}\left(\text{evalc}\left(e^{\omega[1]_2 t} t2[1]\right)\right)$$

$$> \text{valos2} := x1(t) = \text{Re}\left(\text{evalc}\left(e^{\omega[2]_1 t} t1[2]\right)\right), x2(t) = \text{Re}\left(\text{evalc}\left(e^{\omega[2]_1 t} t2[2]\right)\right)$$

$$> \text{kepzetes2} := x1(t) = \text{Im}\left(\text{evalc}\left(e^{\omega[2]_1 t} t1[2]\right)\right), x2(t) = \text{Im}\left(\text{evalc}\left(e^{\omega[2]_1 t} t2[2]\right)\right)$$

$$\text{valos1} := x1(t\sim) = \frac{1}{2} \cos(t\sim), x2(t\sim) = \cos(t\sim)$$

$$\text{kepzetes1} := x1(t\sim) = -\frac{1}{2} \sin(t\sim), x2(t\sim) = -\sin(t\sim)$$

$$\text{valos2} := x1(t\sim) = -2 \cos(\sqrt{6} t\sim), x2(t\sim) = \cos(\sqrt{6} t\sim)$$

$$\text{kepzetes2} := x1(t\sim) = -2 \sin(\sqrt{6} t\sim), x2(t\sim) = \sin(\sqrt{6} t\sim)$$

Ellenőrizzük, hogy a valós függvények valóban megoldásai a differenciálegyenletrendszernek.

> *subs(valos1, de1) : simplify(%); subs(valos1, de2) : simplify(%); subs(kepzetes1, de1) : simplify(%); subs(kepzetes1, de2) : simplify(%)*

$$-\frac{1}{2} \cos(t\sim) = -\frac{1}{2} \cos(t\sim)$$

$$-\cos(t\sim) = -\cos(t\sim)$$

$$\frac{1}{2} \sin(t\sim) = \frac{1}{2} \sin(t\sim)$$

$$\sin(t\sim) = \sin(t\sim)$$

> *subs(valos2, de1) : simplify(%); subs(valos2, de2) : simplify(%); subs(kepzetes2, de1) : simplify(%); subs(kepzetes2, de2) : simplify(%)*

$$12 \cos(\sqrt{6} t\sim) = 12 \cos(\sqrt{6} t\sim)$$

$$-6 \cos(\sqrt{6} t\sim) = -6 \cos(\sqrt{6} t\sim)$$

$$12 \sin(\sqrt{6} t\sim) = 12 \sin(\sqrt{6} t\sim)$$

$$-6 \sin(\sqrt{6} t\sim) = -6 \sin(\sqrt{6} t\sim)$$

Tehát a kapott 4 darab valós értékű függvény a differenciálegyenletrendszer egy-egy partikuláris megoldása. Ezek lineáris kombinációja adja a rendszer általános megoldását.

A Maple *dsolve* utasításával közvetlenül is megkapjuk az általános megoldást.

> *dsolve({de1, de2}, {x1(t), x2(t)})*

$$\left\{ x1(t\sim) = _C1 \sin(t\sim) + _C2 \cos(t\sim) + _C3 \sin(\sqrt{6} t\sim) + _C4 \cos(\sqrt{6} t\sim), x2(t\sim) \right.$$

$$= 2_C1 \sin(t\sim) + 2_C2 \cos(t\sim) - \frac{1}{2} _C3 \sin(\sqrt{6} t\sim) - \frac{1}{2} _C4 \cos(\sqrt{6} t\sim) \}$$

Legyenek a kezdeti feltételek

$$> kezdeti := x1(0) = 1, x2(0) = 2, D(x1)(0) = -2\sqrt{6}, D(x2)(0) = \sqrt{6}$$


$$kezdeti := x1(0) = 1, x2(0) = 2, D(x1)(0) = -2\sqrt{6}, D(x2)(0) = \sqrt{6}$$

$$> mo := dsolve(\{de1, de2, kezdeti\}, \{x1(t), x2(t)\})$$

$$mo := \{x1(t\sim) = \cos(t\sim) - 2 \sin(\sqrt{6} t\sim), x2(t\sim) = 2 \cos(t\sim) + \sin(\sqrt{6} t\sim)\}$$

Ábrázoljuk az adott kezdeti feltételeket kielégítő megoldásokat.

$$> plot([rhs(mo_1), rhs(mo_2)], t=0..3\pi, color=[blue, red])$$



▼ 5. 11. Feladatok

1. feladat Határozzuk meg az adott mátrixok sajátértékeit és sajátvektorait..

$$1. a. \quad \mathbf{A} = \begin{bmatrix} 2 & 0 & -3 \\ 0 & 2 & 3 \\ -3 & 3 & -2 \end{bmatrix}$$

$$1. b. \quad \mathbf{B} = \begin{bmatrix} 1 & 0 & 2 \\ 0 & 1 & 0 \\ 0 & 0 & 2 \end{bmatrix}$$

2. feladat Mutassuk meg, hogy az $\mathbf{A} := \begin{bmatrix} 3 & 2 & 2 \\ 2 & 2 & 0 \\ 2 & 0 & 4 \end{bmatrix}$ szimmetrikus mátrix sajátvektorai

ortogonálisak!

3. feladat Adjuk meg a következő mátrixok legnagyobb abszolút értékű sajátértékeit hatványmódszerrel.

$$3. a. \quad \mathbf{A} = \begin{bmatrix} 2 & 0 & 0 & 0 \\ 2 & 1 & 0 & 0 \\ 2 & 1 & 2 & 0 \\ 2 & 1 & 2 & 1 \end{bmatrix}$$

$$3. b. \quad \mathbf{B} = \begin{bmatrix} 0 & 0 & 1 & 1 \\ 0 & 1 & 0 & 1 \\ 1 & 0 & 1 & 0 \\ 1 & 1 & 0 & 0 \end{bmatrix}$$

4. feladat Transzformáljuk kanonikus alakra a $2xy - 6yz + 2xz$ kvadratikus alakot.

▼ 5. 12. Ellenőrző kérdések

1. Mit értünk mátrix sajátértéke és sajátvektora alatt?
2. Mi a sajátvektor létezésének szükséges és elégséges feltétele?
3. Fogalmazza meg Gersgorin tételét!
- 4; Írja le, milyen elven közelíti a legnagyobb sajátértéket a hatványmódszer!
- 5; Mondja ki a főtengetlytételt!
- 6; Másodfokú kifejezéseket hogyan lehet kanonikus alakra hozni?